```
> restart:
> with(plots):
> # let's write what f is, and the two right-hand sides of the ODE
> f(x,y):=(x*y^g)/(x*y^g+y^g+1);
```

$$f(x, y) := \frac{x\, y^g}{x\, y^g + y^g + 1}$$

```
> eq1:=nu-f(x,y);
```

$$eq1 := \nu - \frac{x\, y^g}{x\, y^g + y^g + 1}$$

```
> eq2:=alpha*f(x,y)-eta*y;
```

$$eq2 := \frac{\alpha\, x\, y^g}{x\, y^g + y^g + 1} - \eta\, y$$

```
> # now let us find the two nullclines
> nul1:=simplify(solve(eq1,y));
```

$$nul1 := \left(-\frac{\nu}{\nu\, x + \nu - x}\right)^{\left(\frac{1}{g}\right)}$$

```
> nul2:=simplify(solve(eq2,y));nul2:=simplify(solve(eq2,y));
```

$$nul2 := \mathrm{RootOf}(\alpha\, x\, \_Z^g - \eta\, \_Z^{(1+g)}\, x - \eta\, \_Z^{(1+g)} - \eta\, \_Z)$$

$$nul2 := \mathrm{RootOf}(\alpha\, x\, \_Z^g - \eta\, \_Z^{(1+g)}\, x - \eta\, \_Z^{(1+g)} - \eta\, \_Z)$$

```
> # let's pick the constants that the book uses, and substitute into
  nullcline equations
> consts:={nu=0.0285,eta=0.1,alpha=1,g=2};
```

$$consts := \{\nu = .0285, \eta = .1, \alpha = 1, g = 2\}$$

```
> nul1:=subs(consts,nul1);
```

$$nul1 := \sqrt{-.0285\, \frac{1}{-.9715\, x + .0285}}$$

```
> nul2:=subs(consts,nul2);
```

$$nul2 := \mathrm{RootOf}(x\, \_Z^2 - .1\, \_Z^3\, x - .1\, \_Z^3 - .1\, \_Z)$$

```
> # this looks messy, but maybe Maple will be able to handle it (?)
> p1:=plot(nul1,x=0..2,thickness=3,color=black):
> p2:=plot(nul2,x=0..2,thickness=3,color=red):
> display(p1,p2);
```




```
> # oops, Maple really messed up with that second graph
> # so maybe it is easier to draw x against y and transpose the
  graphs?  Let us see:
> nul2transpose:=simplify(solve(eq2,x));
```

$$nul2transpose := -\frac{\eta\, y\, (1 + y^{(-g)})}{-\alpha + \eta\, y}$$

```
> nul2transpose:=subs(consts,nul2transpose);
```

$$nul2transpose := -.1\, \frac{y\left(1 + \frac{1}{y^2}\right)}{-1 + .1\, y}$$

```
> p2:=plot([nul2transpose,y,y=0..2]):
> display(p1,p2);
```




```
> # OK, that's better!
> # but where is that steady state, exactly?  Looks like (0.4,0.3)
  or so...
> temp1:=simplify(solve({eq1,eq2},{x,y}));
```

$$temp1 := \left\{ x = -\frac{\nu \left(1 + \left(\frac{\alpha\,\nu}{\eta}\right)^{(-g)}\right)}{\nu - 1}, y = \frac{\alpha\,\nu}{\eta} \right\}$$

```
> X:=subs(temp1,x);
```

$$X := -\frac{\nu\left(1+\left(\frac{\alpha\,\nu}{\eta}\right)^{(-g)}\right)}{\nu-1}$$

```
> Y:=subs(temp1,y);
```

$$Y := \frac{\alpha\,\nu}{\eta}$$

```
> (subs(consts,X),subs(consts,Y));
```

$$.3905066321,\ .2850000000$$

```
> # we were close  :-)
> # btw, the book writes everything in terms of 'p':
> subs(alpha*nu/eta=p,Y);
```

$$p$$

```
> subs(alpha*nu/eta=p,X);
```

$$-\frac{\nu\,(1+p^{(-g)})}{\nu-1}$$

```
> # what about the directions of flow in each of the regions?
> # when x~0, we have that dx/dt is what sign?
> eq1;
```

$$\nu-\frac{x\,y^g}{x\,y^g+y^g+1}$$

```
> subs(x=0,eq1);
```

$$\nu$$

```
> # so it is positive there; and as x goes to infty, we get nu-1<0
  (nu<1 assumed)
> # so we have these signs for dx/dt (direction can only change on
  nullclines)
> with(plottools):
> arrow1 := arrow([0.9,0.6],[0.6,0.6], .1,.2,.1, color=green):
> arrow2 := arrow([0.5,0.13],[0.7,0.13], .1,.2,.1, color=green):
> arrow2b := arrow([0.05,0.3],[0.25,0.3], .1,.2,.1, color=green):
> arrow1b:= arrow([0.4,1.5],[0.1,1.5], .1,.2,.1, color=green):
> display(p1,arrow1,arrow2,arrow1b,arrow2b);
```




```
> # when x~0, we have that dy/dt is what sign?
> eq2;
```

$$\frac{\alpha x y^g}{x y^g + y^g + 1} - \eta y$$

```
> subs(x=0,eq2);
```

$$-\eta y$$

```
> # so negative; while for x -> infty: "a-by" which is positive for
  y small
> # and remember that direction can only change on the nullclines,
  by definition!
```

```
> arrow3 := arrow([0.1,1],[0.1,0.6], .1,.2,.1, color=green):
> arrow4 := arrow([0.9,0.15],[0.9,0.55], .1,.2,.1, color=green):
> display(p2,arrow3,arrow4);
```




```
> # so let'sput everything together now:
> display(p1,p2,arrow1b,arrow2,arrow3,arrow4);
```




```
> # so this gives the directions of motion in the 4 regions
> arrownw := arrow([0.8,0.5],[0.5,0.8], .05,.1,.1, color=green):
> arrowse := arrow([0.1,0.4],[0.4,0.1], .05,.1,.1, color=green):
> arrowsw := arrow([0.25,2],[0.05,1.7], .05,.1,.1, color=green):
> arrowne := arrow([0.9,0.1],[1,0.2], .05,.1,.1, color=green):
> display(p1,p2,arrowsw,arrowse,arrownw,arrowne);
```




```
> # looks like it could be a spiral or a periodic orbit
>
> # *** SEE SEPARATE FILE FOR TRAPPING REGION ARGUMENT !
>
> # so there is a trapping region - we expect Poincare-Bendixon
  might help...
> # let's first sketch a phase plane and solution with Maple
> with(DEtools):
Warning, new definition for translate
> rhs1:=subs(consts,subs({x=x(t),y=y(t)},eq1));
```

$$rhs1 := .0285 - \frac{x(t)\, y(t)^2}{x(t)\, y(t)^2 + y(t)^2 + 1}$$

```
> rhs2:=subs(consts,subs({x=x(t),y=y(t)},eq2));
```

$$rhs2 := \frac{x(t)\, y(t)^2}{x(t)\, y(t)^2 + y(t)^2 + 1} - .1\, y(t)$$

```
> eqn:=diff(x(t),t)=rhs1,diff(y(t),t)=rhs2;
```

$$eqn := \frac{\partial}{\partial t} x(t) = .0285 - \frac{x(t)\, y(t)^2}{x(t)\, y(t)^2 + y(t)^2 + 1}, \frac{\partial}{\partial t} y(t) = \frac{x(t)\, y(t)^2}{x(t)\, y(t)^2 + y(t)^2 + 1} - .1\, y(t)$$

```
> init:=[[x(0)=0.3,y(0)=0.3],[x(0)=1.4,y(0)=0.6]];
```

$$init := [[x(0) = .3, y(0) = .3], [x(0) = 1.4, y(0) = .6]]$$

```
> DEplot([eqn],[x(t),y(t)],t=0..1000,init,x=0..1.5,y=0..1.5,lineco
lor=[black,blue],stepsize=0.1);
```




```
> # so we see a limit cycle, approached from inside (blue) and
  outside
> # let's plot the blue trajectory as a function of t:
> sol:=dsolve({eqn,x(0)=0.3,y(0)=0.3},{x(t),y(t)},type=numeric):
> odeplot(sol,[[t,x(t)],[t,y(t)]],0..1000,numpoints=1000);
```




```
> # now let us study the problem more theoretically
> with(linalg):
Warning, new definition for adjoint
Warning, new definition for norm
Warning, new definition for trace
> F:=vector([eq1,eq2]);
```

$$F := \left[\nu - \frac{x\,y^g}{x\,y^g + y^g + 1},\ \frac{\alpha\,x\,y^g}{x\,y^g + y^g + 1} - \eta\,y \right]$$

```
> Jac:=simplify(jacobian(F, [x,y]));
```

$Jac :=$

$$\left[\begin{array}{cc} -\frac{y^{(2g)}+y^g}{(x\,y^g+y^g+1)^2} & -\frac{x\,g\,y^{(g-1)}}{(x\,y^g+y^g+1)^2} \\ \frac{\alpha\,(y^{(2g)}+y^g)}{(x\,y^g+y^g+1)^2} & \frac{\alpha\,x\,y^{(g-1)}\,g-\eta\,x^2\,y^{(2g)}-2\,\eta\,x\,y^{(2g)}-2\,\eta\,x\,y^g-\eta\,y^{(2g)}-2\,\eta\,y^g-\eta}{(x\,y^g+y^g+1)^2} \end{array}\right]$$

```
> tr:=simplify(trace(Jac));
```

$$tr := \frac{-y^{(2g)}-y^g+\alpha\,x\,y^{(g-1)}\,g-\eta\,x^2\,y^{(2g)}-2\,\eta\,x\,y^{(2g)}-2\,\eta\,x\,y^g-\eta\,y^{(2g)}-2\,\eta\,y^g-\eta}{(x\,y^g+y^g+1)^2}$$

```
> tr0:=simplify(subs(consts,subs({x=X,y=Y},tr)));
```

$$tr0 := .008801452057$$

```
> #so the trace is negative at the equilibrium, when using the above
  params
> determ:=simplify(det(Jac));
```

$$determ := \frac{\eta\,(y^{(4g)}\,x^2+2\,y^{(4g)}\,x+4\,y^{(3g)}\,x+y^{(4g)}+3\,y^{(3g)}+3\,y^{(2g)}+x^2\,y^{(3g)}+2\,x\,y^{(2g)}+y^g)}{(x\,y^g+y^g+1)^4}$$

```
> det0:=simplify(subs(consts,subs({x=X,y=Y},determ)));
```

$$det0 := .007090212492$$

```
> # and the det is positive - so we have an UNSTABLE equilibrium
> # the discriminant is:
> tr0^2-4*det0;
```

$$-.02828338441$$

```
> # negative, SO UNSTABLE SPIRAL
> # together with the "trapping region" argument, we know there's a
  periodic orbit
> # which Maple already indicated too... it is really a limit cycle
> # how does this all depend on the actual parameters???
> # the DET is always > 0, as is clear from the formula above
  (recall x, etc, all >0)
> # let's see what happens to the trace as we vary 'nu'
> # (non-dimensionalized rate of production of ATP, in 0..1 range):
> trac:=simplify(subs({x=X,y=Y},tr));
```

$$trac := -\left(\left(\frac{\alpha\,\nu}{\eta}\right)^{(2g)}+\left(\frac{\alpha\,\nu}{\eta}\right)^g-2\,\nu\left(\frac{\alpha\,\nu}{\eta}\right)^g-g\,\eta+g\left(\frac{\alpha\,\nu}{\eta}\right)^g\eta\,\nu+g\,\eta\,\nu-g\left(\frac{\alpha\,\nu}{\eta}\right)^g\eta+\eta\left(\frac{\alpha\,\nu}{\eta}\right)^{(2g)}+2\,\eta\left(\frac{\alpha\,\nu}{\eta}\right)^g+\left(\frac{\alpha\,\nu}{\eta}\right)^{(2g)}\nu^2-2\left(\frac{\alpha\,\nu}{\eta}\right)^{(2g)}\nu+\left(\frac{\alpha\,\nu}{\eta}\right)^g\nu^2+\eta\right)\Bigg/\left(\left(\frac{\alpha\,\nu}{\eta}\right)^g+1\right)^2$$

```
> factor(subs(nu=0,trac));
```

$$\eta (g - 1)$$

```
> factor(subs(nu=1,trac));
```

$$-\frac{\eta\left(\left(\frac{\alpha}{\eta}\right)^{(2\,g)} + 2\left(\frac{\alpha}{\eta}\right)^{g} + 1\right)}{\left(\left(\frac{\alpha}{\eta}\right)^{g} + 1\right)^{2}}$$

```
> simplify(%+eta);
```

$$0$$

```
> # (sometimes Maple needs to be helped along :)  so it equals -eta
> # so, assuming that g>1 (cooperativity), we see that the trace
> # (evaluated at the steady state, of course) is positive when nu=0
  (instability),
> # and it is negative (stability) when the rate of production of
  ATP is high
>
```